/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

package org.grails.datastore.gorm.events

import groovy.transform.CompileStatic

import org.springframework.context.ApplicationEvent
import org.springframework.context.ApplicationListener
import org.springframework.context.PayloadApplicationEvent
import org.springframework.context.event.SmartApplicationListener

/**
 * Simple implementation that just iterates over a sequence of listeners
 *
 * @author Graeme Rocher
 * @since 6.0
 */
@CompileStatic
class DefaultApplicationEventPublisher implements ConfigurableApplicationEventPublisher {

    private List<ApplicationListener> applicationListeners = []
    @Override
    void publishEvent(ApplicationEvent event) {
        for (listener in applicationListeners) {
            if (listener instanceof SmartApplicationListener) {
                SmartApplicationListener smartApplicationListener = (SmartApplicationListener) listener
                if (!smartApplicationListener.supportsEventType((Class<ApplicationEvent>) event.getClass())) {
                    continue
                }
                else if (!smartApplicationListener.supportsSourceType(event.source.getClass())) {
                    continue
                }
            }
            listener.onApplicationEvent(event)
        }
    }

    @Override
    void publishEvent(Object event) {
        for (listener in applicationListeners) {
            def eventObject = new PayloadApplicationEvent<Object>(this, event)
            if (listener instanceof SmartApplicationListener) {
                SmartApplicationListener smartApplicationListener = (SmartApplicationListener) listener
                if (!smartApplicationListener.supportsEventType((Class<ApplicationEvent>) eventObject.getClass())) {
                    continue
                }
                else if (!smartApplicationListener.supportsSourceType(eventObject.source.getClass())) {
                    continue
                }
            }

            listener.onApplicationEvent(eventObject)
        }
    }

    @Override
    void addApplicationListener(ApplicationListener<?> listener) {
        applicationListeners.add(listener)
    }
}
